/*
 * Copyright 2011 the original author or authors.
 *
 * Licensed under the Apache License, Version 2.0 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 *
 *      http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */
package org.gradle.plugins.signing.type

import org.gradle.plugins.signing.signatory.Signatory

abstract class AbstractSignatureType implements SignatureType {

    File sign(Signatory signatory, File toSign) {
        def signatureFile = fileFor(toSign)
        toSign.withInputStream { toSignStream ->
            signatureFile.withOutputStream { signatureFileStream ->
                sign(signatory, toSignStream, signatureFileStream)
            }
        }
        signatureFile
    }
    
    void sign(Signatory signatory, InputStream toSign, OutputStream destination) {
        signatory.sign(toSign, destination)
    }

    File fileFor(File toSign) {
        new File(toSign.path + ".${getExtension()}")
    }
    
    String combinedExtension(File toSign) {
        def name = toSign.name
        def dotIndex = name.lastIndexOf(".")
        if (dotIndex == -1 || dotIndex + 1 == name.size()) {
            getExtension()
        } else {
            name[++dotIndex..-1] + ".${getExtension()}"
        }
    }
    
}